
Abstract 

Web-Based Enterprise Management (WBEM) is a set of management and Internet standard technologies 
developed to unify the management of enterprise computing environments. WBEM provides the ability for 
the industry to deliver a well-integrated set of standard-based management tools leveraging the emerging 
Web technologies. The DMTF has developed a core set of standards that make up WBEM, which 
includes a data model, the Common Information Model (CIM) standard; an encoding specification, 
Representation of CIM using XML specification; and a transport mechanism, CIM Operations over HTTP.  

This document defines the Universal Resource Identifier (URI) format for WBEM protocols.  A WBEM URI 
is a compact string of characters for identifying a CIM element.  This document defines a mapping of CIM 
naming as defined in the CIM Specification to the URI syntax as defined in RFC 3986. 

The WBEM URI will be used by WBEM protocols as the method for identifying CIM Elements. 

Notice 

DSP0207        Status:  Preliminary 

Copyright © 2004 Distributed Management Task Force, Inc. (DMTF). All rights reserved. 
DMTF is a not-for-profit association of industry members dedicated to promoting enterprise and systems 
management and interoperability. Members and non-members may reproduce DMTF specifications and 
documents for uses consistent with this purpose, provided that correct attribution is given.  As DMTF 
specifications may be revised from time to time, the particular version and release date should always be 
noted. 
Implementation of certain elements of this standard or proposed standard may be subject to third party 
patent rights, including provisional patent rights (herein "patent rights"). DMTF makes no representations 
to users of the standard as to the existence of such rights, and is not responsible to recognize, disclose, 
or identify any or all such third party patent right, owners or claimants, nor for any incomplete or 
inaccurate identification or disclosure of such rights, owners or claimants. DMTF shall have no liability to 
any party, in any manner or circumstance, under any legal theory whatsoever, for failure to recognize, 
disclose, or identify any such third party patent rights, or for such party’s reliance on the standard or 
incorporation thereof in its product, protocols or testing procedures. DMTF shall have no liability to any 
party implementing such standard, whether such implementation is foreseeable or not, nor to any patent 
owner or claimant, and shall have no liability or responsibility for costs or losses incurred if a standard is 
withdrawn or modified after publication, and shall be indemnified and held harmless by any party 
implementing the standard from any and all claims of infringement by a patent owner for such 
implementations. 

Foreword 

The Distributed Management Task Force, Inc. (DMTF), developer of CIM, is the industry organization 
leading the development, adoption, and interoperability of management specifications and initiatives for 
enterprise and Internet environments. 

This specification was processed and approved by the DMTF.  Approval of this specification does not 
necessarily imply that all members voted for approval. 

Introduction 

This specification defines the WBEM Universal Resource Identifier (URI).  The WBEM URI is used in 
WBEM protocols to identify several kinds of CIM objects. 

A URI (as defined by the IETF in RFC 3986) is a compact string representation for a resource available 
via the Internet. This specification defines the subset of the URI syntax that is specific to WBEM. 

The reader is expected to have a working knowledge of URI and WBEM. 

The WBEM URI specification is divided into the following clauses: 

- Clause 1 is the scope; 

- Clause 2 enumerates the normative references that apply to this specification; 

- Clause 3 describes the definitions, symbols, conventions and abbreviations used in this specification; 

- Clause 4 describes the WBEM URI. 

Annexes A through B are for information purposes only. 



WBEM URI Mapping Specification DSP0207 1.0.0 

February 9, 2006  7 
 
 
 
 
 
 
 
 

1 Scope 

This specification defines a URI used by WBEM Protocols as a mapping between CIM Naming, as 
defined in the CIM Infrastructure Specification (DSP0004), and the URI format. 

The URI Generic Syntax standard and the Universal Resource Locators standards by the IETF provide a 
framework for identifying resources. 

2 References 

1. "Common Information Model (CIM) Infrastructure Specification", Version 2.3 DSP0004, DMTF, 4th 
October 2005. 
http://www.dmtf.org/standards/published_documents/DSP0004V2.3_final.pdf 

2. “CIM Operations over HTTP” Version 1.1, DSP0200, DMTF, January 2003. 
http://www.dmtf.org/standards/documents/WBEM/DSP200.html 

3. "Uniform Resource Identifiers (URI): Generic Syntax", IETF RFC 3986, January 2005 
(http://www.ietf.org/rfc/rfc3986.txt) 

4. “Guidelines for new URL Schemes”, IETF RFC 2718, November 1999, 
(http://www.ietf.org/rfc/rfc2718.txt) 

5. “Registration Procedures for URL Scheme Names”, IETF RFC 2717, November 1999, 
(http://www.ietf.org/rfc/rfc2717.txt) 

6. “Domain Names - Concepts and Facilities”, IETF RFC 1034, November 1987, 
(http://www.ietf.org/rfc/rfc1034.txt) 

7. “Requirements for Internet Hosts - Applications and Support”, IETF RFC 1123, October 1989, 
(http://www.ietf.org/rfc/rfc1123.txt) 

8. "Key Words for Use in RFCs to Indicate Requirement Levels", BCP 14, RFC 2119, March 1997 
(http://www.ietf.org/rfc/rfc2119.txt) 

9.  "IP Version 6 Addressing Architecture", IETF RFC 2373, July 1998, 
http://www.ietf.org/rfc/rfc2373.txt?number=2373 

10. "Augmented BNF for Syntax Specifications: ABNF", IETF RFC 2234, November 1997, 
http://www.ietf.org/rfc/rfc2234.txt 

3 Definitions, symbols, abbreviations, and conventions 

The key words "MUST", "MUST NOT", "REQUIRED", "SHALL", "SHALL NOT", "SHOULD", "SHOULD 
NOT", "RECOMMENDED", "MAY", and "OPTIONAL" in this document are to be interpreted as described 
in RFC 2119. 

4 WBEM URI  

This section specifies the WBEM URI. 

The WBEM URI can be used to reference the following kinds of CIM objects: 

¥ CIM namespaces 

¥ CIM classes 

¥ CIM instances 

¥ CIM qualifier types 

There are two formats defined in this version of the WBEM URI. The Untyped WBEM URI is included to 
support legacy use of the ObjectName syntax defined in DMTF DSP0004, CIM Infrastructure 
Specification.  For some applications, the Untyped WBEM URI encoding lacks sufficient type information 
to allow the URI to be processed unambiguously.  For this reason, use of the Typed WBEM URI is 
preferred over the Untyped WBEM URI. 

¥ Untyped WBEM URI – A format compatible with the CIM Object Name format defined by the CIM 
Infrastructure Specification, DSP0004. 

¥ Typed WBEM URI – A format that includes the data type of the key properties, as well as the kind 
of CIM object referenced. 

The WBEM URI consists of the components defined in the following subclauses. Specifications that use 
the WBEM URI MUST define the components that are required in their usage of the WBEM URI. The 
components SHOULD be referenced using the BNF production name defined in this document. 

Figure 1 illustrates the Mapping of the CIM Object Name to the IETF URI. 

Figure 1  WBEM URI as a mapping of CIM Object Name to IETF URI 

4.1 Namespace Type or Scheme 

The Namespace Type is defined as <scheme> in section 3.1 of "Uniform Resource Identifiers (URI): 
Generic Syntax", IETF RFC 3986. 
 
Each WBEM Protocol Specification MUST define the namespace type(s) it supports. 

WBEM defined namespace types MUST be in the form of <protocol>.wbem or <protocol>.wbems where 
protocol is the name of the WBEM protocol. The <protocol>.wbems” MUST be used for the secure 
version of the WBEM protocol. For example, for CIM-XML, the protocol would be cimxml.wbem and 
cimxml.wbems, respectively. A WBEM protocol MAY also support additional namespace types. For 
example, the namespace type “http” and/or “https” would also be valid namespace types for the CIM-XML 
protocol.  

4.2 Authority 

Refer to Section 3.2 of RFC 3986 - Uniform Resource Identifiers (URI): Generic Syntax for the definition 
of the authority. 

4.3 Namespace Name 

The <namespace> MAY contain the “/” character but MUST NOT begin or end with a “/”.  

4.4 Model path encoding 

Two model path encodings are defined in this version of the specification, untyped_modelpath and 
typed_modelpath.   

If the reference is to a CIM Namespace, the model path encoding MUST NOT be present.  
If the reference is to a CIM Class, the model path encoding MUST only include the className.  
If the reference is to an instance, the model path encoding MUST include both the className and a 
key_value_pair or a typed_key_value_pair for all key properties of the instance. 

If the reference is to a Qualifier Type, the model path MUST only include the qualifierName. 

4.4.1 Examples of Untyped Key value pairs 

¥ stringProperty ="This is a string value with a double quote \" included in the string." 
¥ char16Property='a' 
¥ char16Property='\x32' 
¥ booleanProperty=TRUE 
¥ integerProperty=1000 
¥ integerProperty=100101B 
¥ datetimeProperty="19980525133015.000000-300" 
¥ datetimeProperty="00000001132312.000000:000" 
¥ CIM_IndicationSubscription.Filter="CIM_IndicationFilter.SystemCreationClassName=\"CIM_Com

puterSystem\",SystemName=\"server001.acme.com\",CreationClassName=\"CIM_IndicationHand
lerCIMXML\",Name=\"Filter01\"",Handler="CIM_IndicationHandlerCIMXML.SystemCreationClass
Name=\"CIM_ComputerSystem\",SystemName=\"server001.acme.com\",CreationClassName=\"
CIM_IndicationHandlerCIMXML\",Name=\"Handler01\"" 

¥ ACME_DoubleQuoteExample.Antecedent="ACME_AntecedentClass.Name =\"Before double 
quote \\\" after double quote\"",ACME_DependentClass.Name=\"Before backslash \\\\ after 
backslash\"" 

 
4.4.2 Examples of Typed Key value pairs 

¥ stringProperty = (string)"This is a string value with a double quote \" included in the string." 
¥ char16Property=(char16)'a' 
¥ char16Property=(char16)'\x32' 
¥ booleanProperty=(boolean)TRUE 
¥ integerProperty=(sint32)1000 
¥ integerProperty=(uint32)100101B 
¥ integerProperty=(sint64)-12310 
¥ datetimeProperty=(datetime)"19980525133015.000000-300" 
¥ datetimeProperty=(datetime)"00000001132312.000000:000" 
¥ CIM_IndicationSubscription.Filter=(reference)"CIM_IndicationFilter.SystemCreationClassName=(

string)\"CIM_ComputerSystem\",SystemName=(string)\"server001.acme.com\",CreationClassNa

me=(string)\"CIM_IndicationHandlerCIMXML\",Name=(string)\"Filter01\"",Handler=(reference)"CI
M_IndicationHandlerCIMXML.SystemCreationClassName=(string)\"CIM_ComputerSystem\",Syst
emName=(string)\"server001.acme.com\",CreationClassName=(string)\"CIM_IndicationHandlerCI
MXML\",Name=(string)\"Handler01\"" 

¥ ACME_DoubleQuoteExample.Antecedent=(reference)"ACME_AntecedentClass.Name 
=(string)\"Before double quote \\\" after double 
quote\"",ACME_DependentClass.Name=(string)\"Before backslash \\\\ after backslash\"" 

 
4.5 Collected BNF for WBEM URI  

1) The DMTF WBEM URI BNF is a conformant subset of the BNF defined in RFC 3986 (Uniform 
Resource Identifiers (URI): Generic Syntax).  To minimize the impact of future changes, 
implementations should be designed to reject, as invalid or unsupported, WBEM URIs that do not 
conform to an approved version of the DMTF WBEM URI BNF.  

2) This BNF conforms to ABNF as specified by RFC2234: "Augmented BNF for Syntax 
Specifications: ABNF" 

3) The defining specification MUST define the allowable character encodings (e.g. UTF-8 or UCS-2) 
for a WBEM URI. 

 
WBEM-URI              = WBEM-URI-TypedPath / 
                        WBEM-URI-UntypedPath 
 
WBEM-URI-TypedPath    = WBEM-URI-TypedNamespacePath / 
                        WBEM-URI-TypedClassPath / 
                        WBEM-URI-TypedInstancePath / 
                        WBEM-URI-TypedQualifierTypePath 
 
WBEM-URI-UntypedPath  = WBEM-URI-UntypedNamespacePath / 
                        WBEM-URI-UntypedClassPath / 
                        WBEM-URI-UntypedInstancePath 
 
// Note: The production rules for className and qualifierName are 
// defined in Appendix A (MOF Syntax Grammar Description) of DMTF  
// DSP0004, CIM Infrastructure Specification. 
 
WBEM-URI-TypedNamespacePath      = namespacePath "/(namespace)" 
WBEM-URI-TypedClassPath          = namespacePath "/(class)" className 
WBEM-URI-TypedInstancePath       = namespacePath "/(instance)"  
                                   className "." typed_key_value_pairs 
WBEM-URI-TypedQualifierTypePath  = namespacePath "/(qualifiertype)"  
                                   qualifierName 
 
WBEM-URI-UntypedNamespacePath    = namespacePath 
WBEM-URI-UntypedClassPath        = namespacePath ":" className 
WBEM-URI-UntypedInstancePath     = namespacePath ":"  
                                   className "." key_value_pairs 
 
namespacePath         = [namespaceType ":"] namespaceHandle 
namespaceType         = ("http" ["s"]) / ("cimxml.wbem" ["s"]) 

// Note: The production rules for authority are defined in IETF  
// RFC 3986 (Uniform Resource Identifiers (URI): Generic Syntax). 
namespaceHandle       = ["//" authority] "/" [namespaceName] 
 
// Note: IDENTIFIER is an identifier for CIM naming as defined in Appendix F 
// (Unicode Usage) of DMTF DSP0004, CIM Infrastructure Specification. 
namespaceName         = IDENTIFIER *("/"IDENTIFIER) 
 
// Note: The production rules for stringValue, charValue, booleanValue,  
// integerValue, and realValue are defined in Appendix A (MOF Syntax  
// Grammar Description) of DMTF DSP0004, CIM Infrastructure Specification. 
// The production datetimeValue is a datetime value as defined in  
// Section 2.2.1 (Datetime Type) of DMTF DSP0004, CIM Infrastructure 
// Specification. 
 
// Untyped key value pairs  
key_value_pairs       = key_value_pair *("," key_value_pair)  
key_value_pair        = key_name "=" key_value 
key_value             = stringValue / charValue / booleanValue /  
                        integerValue / realValue /  
                        "\"" datetimeValue "\"" / 
                        "\"" referenceValue "\"" 
 
// Typed key value pairs 
typed_key_value_pairs = typed_key_value_pair *("," typed_key_value_pair)  
typed_key_value_pair  = key_name "=" typed_key_value 
typed_key_value       = typed_string_value / typed_char_value / 
                        typed_integer_value / typed_boolean_value / 
                        typed_datetime_value / typed_real_value / 
                        typed_reference_value 
typed_string_value    = "(string)" stringValue 
typed_char_value      = "(char16)" charValue 
typed_boolean_value   = "(boolean)" booleanValue 
typed_integer_value   = ("(uint8)" / "(sint8)" /"(uint16)"/ "(sint16)"/ 
                        "(uint32)" / "(sint32)" / "(uint64)" / 
                        "(sint64)") integerValue 
typed_real_value      = ("(real32)" / "(real32)") realValue 
typed_datetime_value  = "(datetime)" "\"" datetimeValue "\"" 
 
// Note: A typed-reference-value MUST consist of a reference-type designation, 
// followed by a referenceValue enclosed in double quotes (").  A 
// referenceValue MUST be constructed by recursively following these rules to 
// construct a reference to either a CIM Class or CIM Instance. Any double  
// quote character or backslash character that appears in the referenceValue 
// MUST be escaped using a preceding backslash (\) character. 
typed_reference_value = "(reference)" "\"" referenceValue "\"" 
 

 
4.6 WBEM URI Examples 

4.6.1 Untyped Model Path 

//www.acme.com/root/cimv2 
//www.acme.com/root/cimv2:CIM_RegisteredProfile 
https://jdd:test@acme.com:5959/cimv2:CIM_RegisteredProfile 

https://jdd:test@acme.com:5959/cimv2:CIM_RegisteredProfile.InstanceID="acme:1" 
 

4.6.2 Typed Model Path 

//www.acme.com/root/cimv2/(namespace) 
https://jdd:test@acme.com:5959/cimv2/(class)CIM_RegisteredProfile 
https://jdd:test@acme.com:5959/cimv2/(instance)CIM_RegisteredProfile.InstanceID=(string)"acme:1" 
https://jdd:test@acme.com:5959/cimv2/(qualifierType)Abstract 

